Metodo di Simpson

Enrico Bertolazzi

Introduzione

Data la funzione $f(x)$ si approssima

$$\int_a^b f(x)\,dx = \frac{h\left(\sum_{i=1}^{\frac{n}{2}} (f(x_{2\,i-2}) + 4\,f(x_{2\,i-1}) + f(x_{2\,i}))\right)}{3} - \frac{h^4\,(b-a)\,D^{(4)}(f)(z)}{180}$$

dove $h = \frac{b-a}{n}$ e $x_i = a + i\,h$

Carica le librerie

```
> restart ;
  with(plots):
Warning, the name changecoords has been redefined
```

Definisce la procedura Simpson

```
> simpson := proc(f,a,b,n)
    local i, xi, h, r1, r2 ;
    h  := evalf((b-a)/n) ;
    r1 := 0 ;
    for i from 1 to n-1 by 2 do
      r1 := evalf(r1 + f(a+i*h), 20) ;
    end do;
    r2 := 0 ;
    for i from 2 to n-2 by 2 do
      r2 := evalf(r2 + f(a+i*h), 20) ;
    end do;
    return evalf(h*(4*r1+2*r2+f(a)+f(b))/3,20) ;
  end proc :
```

Procedura di Stampa

```
> simpson_print := proc(f,a,b,n)
```

```
    local i, x, X, Y, p, x0, x1, x2, h, GF, GP, C1, C2, C3 ;
    h  := evalf((b-a)/n) ;

    GP := [] ;
    C1 := color=yellow ;
    C2 := color=red ;
    for i from 1 to n-1 by 2 do
      x0 := a + (i-1) * h ;
      x1 := a +      i * h ;
      x2 := a + (i+1) * h ;
      X  := [ x0, x1, x2 ] ;
      Y  := [ evalf(f(x0)), evalf(f(x1)), evalf(f(x2)) ] ;
      p  := interp( X, Y, x ) ;
      GP := [op(GP), plot(p,x=x0..x2,C1,filled=true)]:
      GP := [op(GP), plot(p,x=x0..x2,C2)]:
      C3 := C1 ; C1 := C2 ; C2 := C3 ;
    end do;

    GF := plot(f,a..b,thickness=3,color=blue):
    display({GF,op(GP)},axes=normal,title='Simpson');
  end proc :
```

Esempio d'uso

```
> # Definisce la funzione da approssimare
  f := x -> x*sin(x) ;
```

$$f := x \to x \sin(x)$$

```
> a := 0 ;
  b := 10 ;
  n := 6 ;
```

$$a := 0$$

$$b := 10$$

$$n := 6$$

```
> # risolve il problema
  simpson(f,a,b,n) ;
```

$$8.506698170536649579\,2$$

```
> # disegna la procedura ``simpson''
  simpson_print(f,a,b,n) ;
```



Esercizio: stimare il numero di intervalli per il metodo di Simpson affinch□ l'errore sia inferiore a 10^(-6)

```
> # calcolo la derivata quarta della funzione
  f4 := (D@@4)(f) ;
```

$$f4 := x \rightarrow -4\cos(x) + x\sin(x)$$

```
> # discegno la derivata nell'intervallo considerato e stimo a vista la
  sua maggiorazione
  plot(f4,a..b) ;
```




```
> # la maggiorazione pu essere |f4(x)| < 10
  f4max := 10 ;
```

```
> # la stima dell'errore sara quindi
  stima := (b-a)*h^4*f4max/180 = 10^(-6) ;
```

$$stima := \frac{1}{18} h^4 \, f4max = \frac{1}{1000000}$$

```
> # sostituisco h=(b-a)/N ;
  stimaN := subs(h=(b-a)/N,stima) ;
```

$$stimaN := \frac{5000 \, f4max}{9 \, N^4} = \frac{1}{1000000}$$

```
> # ricavo N^4 dalla stima
  N4 := isolate(stimaN,N^4) ;
```

$$N4 := N^4 = \frac{5000000000}{9} f4max$$

```
> # stimo gli intervalli
  evalf(rhs(N4)^(1/4),5) ;
```

$$153.52 \, f4max^{(1/4)}$$

```
> # gli intervalli saranno quindi 274. Controllo!
  evalf(int(f(x),x=a..b)-simpson(f,a,b,274)) ;
```

$$-1.06 \, 10^{-7}$$